

SUMPRODUCT

提供: ExcelPedia

概要

配列同士の積和を求める数学/三角関数?です。

書式

次の書式で引数を指定します。

SUMPRODUCT(配列1, 配列2, ..., 配列255)

各引数の説明

配列 1

計算の対象となる配列を指定します。セル範囲、および、配列定数を指定することが出来ます。

配列 2 . . .

第2引数以降は省略できます。指定できるのは配列 1 に同じ。

- 引数に指定したそれぞれの配列の行数と列数が異なる場合、エラー値 #VALUE!が表示されます。
- 配列に含まれる数値以外の配列要素は 0 とみなされます。

使用例

使用例 1

下記は数値配列の積和を計算しています。

SUMPRODUCT関数使用例

	1 A	B	2 C	D	=SUMPRODUCT(A1:A4,C1:C4)
1	3		5		
	0		8		
2	4		-1		
3					

4					
5					
6					

表のセル部分のみコピーし、ワークシートに貼り付けて、試すことができます。
Internet Explorer 普通に貼り付け / **Firefox** [形式を選択して貼り付け]-[テキスト]

解説

上記の使用例では次のような計算が行われています。
 配列の要素同士の積算を行い、最後にその和を求めています。

`=SUMPRODUCT(A1:A4, C1:C4)`

`=(A1*C1)+(A2*C2)+(A3*C3)+(A4*C4)`

`=(1*2)+(3*5)+(0*8)+(4*-1)`

`=2+15+0-4`

`=13`

使用例 2

下記はSUMPRODUCT関数の配列計算機能を利用した、複数条件での集計の例です。

SUMPRODUCT関数使用例

	A	B	C	D	E
1	A	い	2		<code>=SUMPRODUCT((A1:A4="A")*(B1:B4="い")*(C1:C4))</code>
2	B	ろ	5		
3	C	は	8		
4	A	に	-1		
5					
6					

表のセル部分のみコピーし、ワークシートに貼り付けて、試すことができます。
Internet Explorer 普通に貼り付け / **Firefox** [形式を選択して貼り付け]-[テキスト]

解説

上記の使用例 2 では次のような計算が行われています。
論理値の配列と数値の配列の積算を行い、最後にその和を求めています。

```
=SUMPRODUCT((A1:A4="A")*(B1:B4="い")*(C1:C4))
```

```
=SUMPRODUCT({TRUE;FALSE;FALSE;TRUE}*{TRUE;FALSE;FALSE;FALSE}*{2;5;8;-1})
```

```
=SUMPRODUCT((TRUE*TRUE*2)+(FALSE*FALSE*5)+(FALSE*FALSE*8)+(TRUE*FALSE*-1))
```

```
=2+0+0+0
```

```
=2
```

論理値の積算は次のように計算されます。

```
=TRUE*TRUE = 1
```

```
=TRUE*FALSE = 0
```

似ていますが、次の式では意図通りに計算されません。

```
=SUMPRODUCT(A1:A4="A",B1:B4="い",C1:C4)
```

これは、「配列に含まれる数値以外の配列要素は 0 とみなされる」という SUMPRODUCT 関数の仕様のため、条件式によって返る論理値の配列がすべて配列の積算の段階で 0 と評価されるためです。

```
=SUMPRODUCT({TRUE;FALSE;FALSE;TRUE},{TRUE;FALSE;FALSE;FALSE},{2;5;8;-1})
```

そのため、「複数条件で集計する」という目的で応用する場合には、論理値として独立した配列としない工夫が必要です。

補足

なお、Excel2007以降では「複数条件での集計」として SUMIFS[?] 関数、「複数条件でのカウント」として COUNTIFS[?] 関数が用意されています。

「<http://localhost/mwiki/index.php/SUMPRODUCT>」より作成

カテゴリ: 数学/三角関数 | 関数

- このページの最終更新は 2013年12月23日 (月) 18:30 に行われました。